<?xml version="1.0" encoding="utf-8"?>
<xsl:stylesheet version="1.0"
     xmlns:xsl="http://www.w3.org/1999/XSL/Transform"
     xmlns:exsl="http://exslt.org/common"
     xmlns:php="http://php.net/xsl"
     xmlns:tpl="http://joos.nnov.ru/xsl"
     xmlns:rml="http://joos.nnov.ru/xml">

  <!-- RES { -->
  <xsl:template match="rml:item" mode="res-article" xmlns="http://www.w3.org/1999/xhtml">
    <tpl:italic>
      <xsl:element name="a">
        <xsl:attribute name="href"><xsl:value-of select="rml:link" /></xsl:attribute>
        <xsl:attribute name="target">_blank</xsl:attribute>
        <xsl:attribute name="class">article_link</xsl:attribute>

        <xsl:value-of select="rml:title" />
      </xsl:element>
    </tpl:italic>
    <div class="item">
      <xsl:value-of select="rml:content" disable-output-escaping="yes" />
      <div class="date">
        <tpl:message><xsl:value-of select="substring(string(rml:created), 6, 6)" /></tpl:message>
        <tpl:entity name="nbsp" /><tpl:entity name="mdash" /><tpl:entity name="nbsp" />
        <tpl:message><xsl:value-of select="substring(string(rml:created), 18, 5)" /></tpl:message>
      </div>
    </div>
  </xsl:template>

  <xsl:template match="rml:resource" mode="res" xmlns="http://www.w3.org/1999/xhtml">
    <div class="articles">
      <xsl:apply-templates select="rml:articles/rml:item" mode="res-article" />
    </div>
  </xsl:template>
  <!-- } RES -->
  
  <!-- HTML { -->
  <xsl:template match="rml:resource" mode="html" xmlns="http://www.w3.org/1999/xhtml">
    <xsl:element name="tpl:page" namespace="http://joos.nnov.ru/xsl">
      <xsl:attribute name="title"><xsl:value-of select="rml:title" /></xsl:attribute>

      <xsl:attribute name="rss">/index.rss</xsl:attribute>
      <xsl:attribute name="xml">/index.xml</xsl:attribute>
      <xsl:attribute name="dump">/index.dump</xsl:attribute>
      <xsl:attribute name="jsonml">/index.jsonml</xsl:attribute>

      <xsl:apply-templates select="." mode="res" />
      <p align="center"><tpl:liru /></p>

      <tpl:document href="joos://localhost/none.xml" />
    </xsl:element>
  </xsl:template>
  <!-- } HTML -->

  <!-- XML { -->
  <xsl:template match="rml:resource" mode="xml">
    <xsl:apply-templates select="." mode="copy" />
  </xsl:template>
  <!-- } XML -->

  <!-- RSS { -->
  <xsl:template match="rml:resource" mode="rss" xmlns="http://backend.userland.com/rss2">
    <rss version="2.0">
      <channel>
        <title><xsl:value-of select="rml:title" /></title>
        <link><xsl:value-of select="rml:link" /></link>

        <xsl:for-each select="rml:articles/rml:item">
          <item>
            <title>
              <xsl:text disable-output-escaping="yes">&lt;![CDATA[</xsl:text>
              <xsl:value-of select="rml:title" />
              <xsl:text disable-output-escaping="yes">]]&gt;</xsl:text>
            </title>
            <link><xsl:value-of select="rml:link" /></link>
            <pubDate><xsl:value-of select="rml:created" /></pubDate>

            <description>
              <xsl:text disable-output-escaping="yes">&lt;![CDATA[</xsl:text>
              <xsl:value-of select="rml:content" disable-output-escaping="yes" />
              <xsl:text disable-output-escaping="yes">]]&gt;</xsl:text>
            </description>
          </item>
        </xsl:for-each>
      </channel>
    </rss>
  </xsl:template>
  <!-- } RSS -->

</xsl:stylesheet>
